﻿<%@language=vbscript codepage=65001%>
<%
Const PurviewLevel=3    '操作权限 1超级用户2高级管理员3信息录入员
'判断管理是否登陆
'**************************************************************
if session("admin")="" then
    response.Redirect "../Admin_Logout.asp"
else
	if session("purview")>PurviewLevel then
		response.write "<br><p align=center><font color='red' size='2'>您没有操作的权限</font></p>"
		response.end
	end if
end if
%>
<!--#include file="../Config/Config.asp"-->
<!--#include file="upfile_class.asp"-->
<%
const upload_type=0   '上传方法：0=无惧无组件上传类，1=FSO上传 2=lyfupload，3=aspupload，4=chinaaspupload

dim upload,oFile,formName,SavePath,filename,fileExt
dim EnableUpload
dim arrUpFileType
dim ranNum
dim msg,FoundErr
msg=""
FoundErr=false
EnableUpload=false
SavePath=SaveUpFilesPath   '存放上传文件的目录
if right(SavePath,1)<>"/" then SavePath=SavePath&"/" '在目录后加(/)
SavePath=SavePath & Year(date()) &"/"& Month(date()) &""& day(date()) &"/"
%>
<html>
<head>
<meta http-equiv="Content-Type" content="text/html; charset=utf-8">
<title>上传</title>
</head>
<body leftmargin="2" topmargin="0" marginwidth="0" marginheight="0">
<%
if EnableUploadFile="No" then
	response.write "系统未开放文件上传功能"
else
	if session("Admin")="33" then
		response.Write("请登录后再使用本功能！")
	else
		select case upload_type
			case 0
				call upload_0()  '使用化境无组件上传类
			case else
				'response.write "本系统未开放插件功能"
				'response.end
		end select
	end if
end if
%>
</body>
</html>
<%
sub upload_0()    '使用化境无组件上传类
	set upload=new upfile_class ''建立上传对象
	upload.GetData(10240000)   '取得上传数据,限制最大上传10M
	if upload.err > 0 then  '如果出错
		select case upload.err
			case 1
				response.write "请先选择你要上传的文件！"
			case 2
				response.write "你上传的文件总大小超出了最大限制（10M）"
		end select
		response.end
	end if
		
	for each formName in upload.file '列出所有上传了的文件
		set ofile=upload.file(formName)  '生成一个文件对象
		if ofile.filesize<100 then
			msg="请先选择你要上传的文件！"
			FoundErr=True
		end if
		if ofile.filesize>(MaxFileSize*1024) then
 			msg="文件大小超过了限制，最大只能上传200K的文件！"
			FoundErr=true
		end if

		fileExt=lcase(ofile.FileExt)
		arrUpFileType=split(UpFileType,"|")
		for i=0 to ubound(arrUpFileType)
			if fileEXT=trim(arrUpFileType(i)) then
				EnableUpload=true
				exit for
			end if
		next
		if fileEXT="asp" or fileEXT="asa" or fileEXT="aspx" then
			EnableUpload=false
		end if
		if EnableUpload=false then
			msg="这种文件类型不允许上传！\n\n只允许上传这几种文件类型：" & UpFileType
			FoundErr=true
		end if		
		if FoundErr<>true then
			randomize
			ranNum=int(900*rnd)+100
			filename=SavePath&year(now)&month(now)&day(now)&hour(now)&minute(now)&second(now)&ranNum&"."&fileExt
			CreateFolder FileName
			ofile.SaveToFile Server.mappath(FileName)   '保存文件
		  	response.write "<script>window.returnValue='"& FileName &"';window.close();</script>"
          Else
		  	response.write "<script>alert('"&msg&"');</script>"					
		end if
		set file=nothing
	next
	set upload=nothing
end sub

Sub CreateFolder(byval paths)
		on error Resume Next
		Dim File,Pathdir
		paths=Server.mappath(paths)
		paths=Replace(paths,"\","/")
		set FSO= CreateObject(UpFileFSO)
		pathdir=split(paths,"/")
		for jt=0 to ubound(pathdir)-1
			file=file & pathdir(jt) &"/"
			If not fso.FolderExists(file) Then FSO.createFolder(file)
		next
End Sub
%>